// File: Locked_token_Azur/Locked_token_4.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.19;

/**
 * Create By AZUR Team.
 * AzurLock built for the Azur ecosystem that you can use on "Azurswap.org"
 *
 *
 * It would be an honor to have you use it.
 ** Sincerely, Azureswap Senior Developer
 */


import "./Context.sol";
import "./IERC20.sol";
import "./SafeERC20.sol";
import "./Ownable.sol";

contract AZURLockedToken is Ownable {
    using Address for address;
    using SafeERC20 for IERC20;

    event AddTokentoLockedPoolEvent(
        address tokenAddress,
        address ownerLocked,
        uint256 startTime,
        uint256 endTime,
        uint256 amountToken,
        uint8 decimals,
        string tokenName,
        string tokenSymbol
    );

    event WithdrawTokenofFree(
        address tokenAddress,
        address ownerLocked,
        uint256 amountToken,
        uint256 withdrawTime,
        uint8 decimals,
        string tokenName,
        string tokenSymbol
    );
    event MakeFreeTokensEvent(
        address tokenAddress,
        address ownerLocked,
        uint256 amountToken,
        uint256 timeFreeTokens,
        uint8 decimals,
        string nameToken,
        string symbolToken
    );

    struct InfoLocked {
        uint256 startTime;
        uint256 endTime;
        uint256 amountLockedTokens;
        uint256 amountFreeTokens;
        uint8 decimals;
        string nameToken;
        string symbolToken;
    }

    mapping(address => mapping(address => InfoLocked)) private EachInfoLocked;
    mapping(address => mapping(address => bool)) private AllPairOfAddress;
    mapping(address => uint256) public NumberOfLockedAddress;
    mapping(address => address[]) LockedOfUser;

    function _isContract(address account) internal view returns (bool) {
        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    modifier notContract() {
        require(!_isContract(msg.sender), "Contract not allowed");
        require(msg.sender == tx.origin, "Proxy contract not allowed");
        _;
    }

    constructor() {}

    function AddTokentoLockedPool(
        uint256 endTime,
        address tokenContract,
        uint256 amount
    ) public notContract returns (uint256, uint256) {
        IERC20 itoken = IERC20(address(tokenContract));
        require(amount > 0, "The value must be greater than 0");
        require(
            endTime > block.timestamp,
            "Time must be longer than the present"
        );
        uint256 allowance = itoken.allowance(msg.sender, address(this));
        require(allowance >= amount, "Check the token allowance");
        return _AddTokentoLockedPool(endTime, itoken, amount);
    }

    function _AddTokentoLockedPool(
        uint256 _endTime,
        IERC20 _itoken,
        uint256 _amount
    ) private returns (uint256, uint256) {
        address _Aitoken = address(_itoken);
        uint256 balancebefore = _itoken.balanceOf(address(this));
        _itoken.safeTransferFrom(_msgSender(), address(this), _amount);
        uint256 balanceafter = _itoken.balanceOf(address(this));
        _amount = balanceafter - balancebefore;

        if (!AllPairOfAddress[msg.sender][_Aitoken]) {
            NumberOfLockedAddress[msg.sender]++;
            LockedOfUser[msg.sender].push(_Aitoken);
            AllPairOfAddress[msg.sender][_Aitoken] = true;
            EachInfoLocked[msg.sender][_Aitoken] = InfoLocked(
                block.timestamp,
                _endTime,
                _amount,
                0,
                _itoken.decimals(),
                _itoken.name(),
                _itoken.symbol()
            );
        } else {
            //
            require(
                _endTime >= EachInfoLocked[msg.sender][_Aitoken].endTime,
                "Time must be After Ended Time"
            );
            EachInfoLocked[msg.sender][_Aitoken].amountLockedTokens =
                (EachInfoLocked[msg.sender][_Aitoken].amountLockedTokens) +
                _amount;
            EachInfoLocked[msg.sender][_Aitoken].endTime = _endTime;
            EachInfoLocked[msg.sender][_Aitoken].startTime = block.timestamp;
        }

        emit AddTokentoLockedPoolEvent(
            _Aitoken,
            msg.sender,
            block.timestamp,
            _endTime,
            _amount,
            _itoken.decimals(),
            _itoken.name(),
            _itoken.symbol()
        );
        return (_endTime, _amount);
    }

    function WithdrawTokenFreeOnTime(
        address tokenAddress,
        uint256 amount
    ) public notContract returns (address, uint256, bool) {
        require(amount > 0, "The value must be greater than 0");
        require(AllPairOfAddress[msg.sender][tokenAddress], "Locked Not Exist");
        require(
            EachInfoLocked[msg.sender][tokenAddress].amountFreeTokens >= amount,
            "Not Enough Free Token in Locked"
        );
        require(
            block.timestamp >= EachInfoLocked[msg.sender][tokenAddress].endTime,
            "Time is Not End"
        );
        return _WithdrawTokenFreeOnTime(tokenAddress, amount);
    }

    function _WithdrawTokenFreeOnTime(
        address _tokenAddress,
        uint256 _amount
    ) private returns (address, uint256, bool) {
        IERC20 itoken = IERC20(_tokenAddress);
        itoken.safeTransfer(msg.sender, _amount);
        EachInfoLocked[msg.sender][_tokenAddress].amountFreeTokens =
            (EachInfoLocked[msg.sender][_tokenAddress].amountFreeTokens) -
            _amount;
        emit WithdrawTokenofFree(
            _tokenAddress,
            msg.sender,
            _amount,
            block.timestamp,
            itoken.decimals(),
            itoken.name(),
            itoken.symbol()
        );
        return (_tokenAddress, _amount, true);
    }

    function MakeFreeTokens(
        address tokenCont,
        uint256 amount
    ) public returns (address, uint256) {
        require(amount > 0, "The value must be greater than 0");
        require(AllPairOfAddress[msg.sender][tokenCont], "Locked Not Exist");
        require(
            block.timestamp > EachInfoLocked[msg.sender][tokenCont].endTime,
            "Time is Not End"
        );
        require(
            EachInfoLocked[msg.sender][tokenCont].amountLockedTokens > 0,
            "Locked Pool is Zero Tokens"
        );
        require(
            amount <= EachInfoLocked[msg.sender][tokenCont].amountLockedTokens,
            "Must be less than the locked value"
        );
        return _MakeFreeTokens(tokenCont, amount);
    }

    function _MakeFreeTokens(
        address _tokenCont,
        uint256 _amount
    ) private returns (address, uint256) {
        EachInfoLocked[msg.sender][_tokenCont].amountFreeTokens =
            (EachInfoLocked[msg.sender][_tokenCont].amountFreeTokens) +
            _amount;
        EachInfoLocked[msg.sender][_tokenCont].amountLockedTokens =
            (EachInfoLocked[msg.sender][_tokenCont].amountLockedTokens) -
            _amount;

        IERC20 _itoken = IERC20(_tokenCont);

        emit MakeFreeTokensEvent(
            _tokenCont,
            msg.sender,
            _amount,
            block.timestamp,
            _itoken.decimals(),
            _itoken.name(),
            _itoken.symbol()
        );
        return (_tokenCont, _amount);
    }

    function HowMuchIsFreeOnLocked(
        address ownerLocked,
        address tokenContract
    ) public view returns (uint256) {
        return EachInfoLocked[ownerLocked][tokenContract].amountFreeTokens;
    }

    function HowMuchIsLockedOnLock(
        address ownerLocked,
        address tokenContract
    ) public view returns (uint256) {
        return EachInfoLocked[ownerLocked][tokenContract].amountLockedTokens;
    }

    function WhenIsEndedLocked(
        address ownerLocked,
        address tokenContract
    ) public view returns (uint256) {
        return EachInfoLocked[ownerLocked][tokenContract].endTime;
    }

    function StartLockedTime(
        address ownerLocked,
        address tokenContract
    ) public view returns (uint256) {
        return EachInfoLocked[ownerLocked][tokenContract].startTime;
    }

    function NameLockedPool(
        address ownerLocked,
        address tokenContract
    ) public view returns (string memory) {
        return EachInfoLocked[ownerLocked][tokenContract].nameToken;
    }

    function symbolLockedPool(
        address ownerLocked,
        address tokenContract
    ) public view returns (string memory) {
        return EachInfoLocked[ownerLocked][tokenContract].symbolToken;
    }

    function DecimalsLockedPool(
        address ownerLocked,
        address tokenContract
    ) public view returns (uint8) {
        return EachInfoLocked[ownerLocked][tokenContract].decimals;
    }

    function LockedPoolInfo(
        address ownerLocked,
        address tokenContract
    ) public view returns (InfoLocked memory) {
        return EachInfoLocked[ownerLocked][tokenContract];
    }

    function TokenisExist(
        address ownerLocked,
        address tokenContract
    ) public view returns (bool) {
        return AllPairOfAddress[ownerLocked][tokenContract];
    }

    function UserPools(
        address ownerLocked
    ) public view returns (address[] memory) {
        return LockedOfUser[ownerLocked];
    }
}

// File: Locked_token_Azur/Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.19;

import './Context.sol';

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _setOwner(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _setOwner(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _setOwner(newOwner);
    }

    function _setOwner(address newOwner) private {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}
// File: Locked_token_Azur/SafeERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.19;

import './Address.sol';
import './IERC20.sol';

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(
            token,
            abi.encodeWithSelector(token.transfer.selector, to, value)
        );
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(
            token,
            abi.encodeWithSelector(token.transferFrom.selector, from, to, value)
        );
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(
            token,
            abi.encodeWithSelector(token.approve.selector, spender, value)
        );
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(
            token,
            abi.encodeWithSelector(
                token.approve.selector,
                spender,
                newAllowance
            )
        );
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(
                oldAllowance >= value,
                "SafeERC20: decreased allowance below zero"
            );
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(
                token,
                abi.encodeWithSelector(
                    token.approve.selector,
                    spender,
                    newAllowance
                )
            );
        }
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(
            data,
            "SafeERC20: low-level call failed"
        );
        if (returndata.length > 0) {
            // Return data is optional
            require(
                abi.decode(returndata, (bool)),
                "SafeERC20: ERC20 operation did not succeed"
            );
        }
    }
}

// File: Locked_token_Azur/IERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.19;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);


    function decimals() external view returns (uint8);

    function name() external view returns (string memory);
    
    function symbol() external view returns (string memory);


    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}
// File: Locked_token_Azur/Context.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.19;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: Locked_token_Azur/Address.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.19;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(
            address(this).balance >= amount,
            "Address: insufficient balance"
        );

        (bool success, ) = recipient.call{value: amount}("");
        require(
            success,
            "Address: unable to send value, recipient may have reverted"
        );
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data)
        internal
        returns (bytes memory)
    {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return
            functionCallWithValue(
                target,
                data,
                value,
                "Address: low-level call with value failed"
            );
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(
            address(this).balance >= value,
            "Address: insufficient balance for call"
        );
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(
            data
        );
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data)
        internal
        view
        returns (bytes memory)
    {
        return
            functionStaticCall(
                target,
                data,
                "Address: low-level static call failed"
            );
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data)
        internal
        returns (bytes memory)
    {
        return
            functionDelegateCall(
                target,
                data,
                "Address: low-level delegate call failed"
            );
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    function _verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) private pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}